The Importance of Being Formal

Martin Henz

January 23, 2013

Traditional Logic

1 Motivation

Our first formal system expresses relationships between classes of things. Each class is represented by a noun naming the class, or by an adjective that describes a property that distinguishes the members of the class from non-members. These nouns or adjectives denoting classes of things are called *categorical terms*, or simply *terms*.

For example, the term `animals` refers to the class of things that contains all animals.

As another example, the term `brave` refers to the class of persons that we consider brave. Note that we are not very concerned with the difficulty to deliberate the attribute of bravery in individual cases. In logic, we do not spend much time arguing whether a particular person, say Socrates, is brave or not. Instead, we assume that it is always possible to univocally attest the given attribute to a given person or not.

Categorical terms constitute the basic unit of meaning in our first deductive system, which is therefore called term logic (sometimes also categorical logic). In the history of philosophy, term logic plays a prominent role, because many arguments that appear in everyday discourse (and also in political statements, philosophical treatises, etc) can be analysed and verified using term logic. In fact, term logic has been investigated extensively by the Greek philosopher Aristotle (384–322 BCE) already, whose treatise *Prior Analytics* is considered the earliest study in formal logic and was widely accepted as the definitive approach to deductive reasoning until the 19^{th}century.

To understand the concerns of term logic, consider the following argument.

Example 1.

> *All humans are mortal.*
> *All Greeks are humans.*
> *Therefore, all Greeks are mortal.*

This argument appears acceptable and consistent with our intuition. But what exactly makes it acceptable, and what makes the following "argument" unacceptable?

Example 2.

All cats are predators.
Some animals are cats.
Therefore, all animals are predators.

Of course, we know that the conclusion of the latter "argument" is false, so we know that there must be something wrong with the argument. However, even if we do not know anything about the subject matter, arguments of the first kind are acceptable. Consider for example:

Example 3.

All slack track systems are caterpillar systems.
All Christie suspension systems are slack track systems.
Therefore, all Christie suspension systems are caterpillar systems.

Even if you have no knowledge of caterpillar suspension systems, the argument appears sound. It appeals to you because of the way the categorical terms are arranged in the argument, rather than the choice of the categorical terms, themselves. Our study of logic focuses on what arguments are acceptable, or *hold*, due to their form alone. In order to do so, it is important to precisely define what an argument consists of, what we mean by *validity*, and what methods we can use to demonstrate validity.

2 Terms and their Semantics

In our reasoning framework, we define categorical terms as members of a particular *data type*, `Term`, whose elements name classes of entities. For example, `animals` may be an element of `Term`, representing all entities which are considered animals.

In parallel with the discussion of the topic, we will introduce a formal system that illustrates the particular style of reasoning, supported by a software system called Coq. Throughout this book, we make use of Coq as a *proof assistant*, helping us formalize arguments and assisting us in the construction of proofs, as we advance through a sequence of more and more complex and powerful logics. In Coq, we can write scripts that define the logic

```
Module TraditionalLogic.
```

In Coq, we can define the type `Thing` as follows.

```
Parameter Thing : Type.
```

For example, we can declare that `Socrates` is a `Thing` by writing:

```
Parameter Socrates : Thing.
```

Now we can define our classes as functions that decide whether a given `Thing` is included in the class or not.

```
Parameter humans : Thing -> Prop.
```

When we apply such a function to a `Thing`, such as `Socrates`, denoted

`humans Socrates`

we will get a `Prop`, which means a *proposition* that either holds or does not hold.

Note that the function `humans` represents a categorical term as introduced earlier. Therefore, we call such functions `Term`:

```
Definition Term := Thing -> Prop.
```

With this definition, we can define other terms easier:

```
Parameter Greeks : Term.
Parameter mortal : Term.
```

In this section, we clarify the relationship between such functions what they represent. A particular meaning $\mathcal{M}$, also called *model*, fixes a universe of discourse, denoted $U^{\mathcal{M}}$, and for every element $t \in \texttt{Term}$, a set $t^{\mathcal{M}}$, where $t^{\mathcal{M}} \subseteq U^{\mathcal{M}}$.[1]

For example, for reasoning about living beings such as cats, humans, Greeks, and so on, we may choose a meaning $\mathcal{M}$ whose universe $U^{\mathcal{M}}$ is the set of all living beings, whose $\texttt{cat}^{\mathcal{M}}$ the set of all cats, whose $\texttt{humans}^{\mathcal{M}}$ the set of all humans, and so on. However, for the same terms, we may consider a different meaning $\mathcal{M}'$ whose universe $U^{\mathcal{M}'}$ is a set of labeled playing cards, whose $\texttt{cat}^{\mathcal{M}'}$ contains those cards that display cats, whose $\texttt{humans}^{\mathcal{M}'}$ contains those cards that display humans, and so on.

[1] More formally, we can define a semantics $\mathcal{M}$ of a set of terms `Term` as a pair $(U, \mathit{interprete})$, where U is a set, and *interprete* is a function

$$\mathit{interprete} : \texttt{Term} \to \mathcal{P}(U)$$

where $\mathcal{P}(U)$ denotes the set of all subsets of U. Thus, for a given `Term` t, $\mathit{interprete}(t)$, denoted by $t^{\mathcal{M}}$, is a subset of U.

Example 4. *Consider the following set of terms:*

$$\mathit{Term} = \{\mathit{even}, \mathit{odd}, \mathit{belowfour}\}$$

We could choose a meaning $\mathcal{M}_1$*, where* $U^{\mathcal{M}_1} = \mathbb{N}$*, and the "obvious" meaning* $\mathit{even}^{\mathcal{M}_1} = \{0, 2, 4, \ldots\}$*,* $\mathit{odd}^{\mathcal{M}_1} = \{1, 3, 5, \ldots\}$*, and* $\mathit{belowfour}^{\mathcal{M}_1} = \{0, 1, 2, 3\}$*.*

Alternatively, we could choose a meaning $\mathcal{M}_2$ *where* $U^{\mathcal{M}_2} = \{0, 1, 2, 3, 4, 5, 6\}$*, and the "obvious" meaning* $\mathit{even}^{\mathcal{M}_2} = \{0, 2, 4, 6\}$*,* $\mathit{odd}^{\mathcal{M}_2} = \{1, 3, 5\}$*, and* $\mathit{belowfour}^{\mathcal{M}_2} = \{0, 1, 2, 3\}$*.*

However, no one can prevent us from choosing an unexpected meaning $\mathcal{M}_3$*, in which* $U^{\mathcal{M}_3} = \{a, b, c, \ldots, z\}$*,* $\mathit{even}^{\mathcal{M}_3} = \{a, e, i, o, u\}$*,* $\mathit{odd}^{\mathcal{M}_3} = \{b, c, d, \ldots\}$*, and* $\mathit{belowfour}^{\mathcal{M}_3} = \emptyset$*.*

3 Categorical Propositions

Our term logic allows us to express relationships between two categorical terms. For example, we may want to investigate the statement

```
All cats are predators
```

This statement expresses a relationship between the terms `cats` and `predators`, saying that every thing that is included in the class represented by `cats` is also included in the class represented by `predators`.

Such statements are called categorical propositions. The first categorical term in the proposition (in our case `cats`) is called the subject of the proposition, and the second term (in our case `predators`) is called its object. Categorical propositions of the form

`All` t_1 `are` t_2

where t_1 and t_2 are terms, are called universal affirmative propositions.

Similarly, we provide for universal negative propositions such as `No Greeks are cats`, particular affirmative propostions such as `Some animals are cats`, and particular negative propositions such as `Some cats are not brave`.

Thus, categorical propositions come in four forms, depending on the *quantity* (universal or particular), and *quality* (affirmative or negative).

Note that all propositions deal with terms that describe the relationship between classes of entities, not individual entities. In order to express that a particular entity, say Socrates, is included in a class, say Greek, one would need to form a term such as "people with the name Socrates", and then state the universal affirmative proposition "All people with the name Socrates are Greek".[2]

[2]Or should it be a particular affirmative proposition "Some people with the name Socrates are Greek"? This question gives you a hint of the philosophical difficulties posed by pushing traditional logic beyond classes of entities.

4 Semantics of Categorical Propositions

Recall that terms represent subsets of a particular domain of discourse. Categorical propositions describe relationships between these sets. For example, the proposition

`All humans are mortal`

says that the set `humans` is a subset of the set `mortal`. Once we fix the sets that the terms represent, it is clear which propositions hold and which don't.

Thus, we can define the meaning of the categorical proposition with respect to a model $\mathcal{M}$ as follows:

$$(\texttt{All}\ \textit{subject}\ \texttt{are}\ \textit{object})^{\mathcal{M}} = \begin{cases} T & \text{if } \textit{subject}^{\mathcal{M}} \subseteq \textit{object}^{\mathcal{M}}, \\ F & \text{otherwise} \end{cases}$$

Here T and F represent the logical truth values *true* and *false*, respectively. We visualize a universal affirmative proposition such as `All Greeks are mortal` using a *Venn diagram* as follows:



The darkest shading indicates an area that does not contain any entities, if the proposition is true.

Alternatively, we can represent the fact that the set $\texttt{Greeks}^{\mathcal{M}}$ is contained in the selt $\texttt{humans}^{\mathcal{M}}$ by the following Venn diagram:



In Coq, we represent universal affirmative propositions by stating that for any `Thing x`, if the subject proposition holds on `x`, then the object proposition also holds on `x`.

```
Definition GreeksMortal : Prop :=
    forall x,
    (Greeks x) -> ( mortal x).
```

In Coq, we can make such a proposition *look like* a categorical proposition by introducing a *notation*:

```
Notation "'All' subject 'are' object " :=
    (forall x, (subject x) -> (object x)) (at level 50).
```

after which we can simply write:

```
Definition GreeksMortal2 : Prop :=
    All Greeks are mortal.
```

Similarly, we can define the meaning of the other categorical propositions:

$$(\texttt{No}\ \mathit{subject}\ \texttt{are}\ \mathit{object})^{\mathcal{M}} = \begin{cases} T & \text{if } \mathit{subject}^{\mathcal{M}} \cap \mathit{object}^{\mathcal{M}} = \emptyset, \\ F & \text{otherwise} \end{cases}$$

Again, the darkest shading in the diagram for `No Greeks are cats` indicates an empty area.



The meaning of particular affirmative propositions is given by

$$(\texttt{Some}\ \mathit{subject}\ \texttt{are}\ \mathit{object})^{\mathcal{M}} = \begin{cases} T & \text{if } \mathit{subject}^{\mathcal{M}} \cap \mathit{object}^{\mathcal{M}} \neq \emptyset, \\ F & \text{otherwise} \end{cases}$$

and visualized through the following diagram.



The darkest region in the diagram for `Some humans are Greeks` now represents an area that contains at least one entity.

Finally, the meaning of particular negative propositions is given by

$$(\texttt{Some}\ \mathit{subject}\ \texttt{are not}\ \mathit{object})^{\mathcal{M}} = \begin{cases} T & \text{if } \mathit{subject}^{\mathcal{M}} / \mathit{object}^{\mathcal{M}} \neq \emptyset, \\ F & \text{otherwise} \end{cases}$$

A proposition such as `Some Greeks are not vegetarians` is visualized by



where the darkest region represents an area that contains at least one entity.

Similarly to universal affirmative propositions, we can introduce notations for the other three kinds of propositions:

```
Notation "'No' subject 'are' object " :=
    (forall x, (subject x) -> ~(object x)) (at level 50).

Notation "'Some' subject 'are' object " :=
    (exists x, (subject x) /\ (object x)) (at level 50).

Notation "'Some' subject 'are' 'not' object " :=
    (exists x, (subject x) /\ ~(object x)) (at level 50).
```

The following examples illustrate the use of these new kinds of propositions.

```
Parameter cats : Term.

Definition NoExample: Prop :=
   No Greeks are cats.

Parameter animals : Term.

Definition SomeExample: Prop :=
   Some animals are cats.

Parameter brave : Term.

Definition SomeNotExample: Prop :=
   Some cats are not brave.
```

5 Axioms, Lemmas and Proofs

We would like to be able to state that a particular categorical proposition holds, so that we can later make use of it as a fact. In logic, propositions that are assumed to hold are called *axioms*.

In order to refer to an axiom later on, we allow ourselves to give it a name. For example, we can state the mortality of humans and the humanity of Greeks as follows.

Axiom 1 (HumansMortality)**.** *The proposition* `All humans are mortal` *holds.*

Axiom 2 (GreeksHumanity)**.** *The proposition* `All Greeks are humans` *holds.*

In Coq, asserting a proposition is done using *axioms* of the form

`Axiom` *name* `:` *proposition*.

Such a *proposition* has the type `Prop` in Coq; it either holds or does not hold.

Now we can assert the mortality of humans and the humanity of Greeks as axioms.

```
Axiom HumansMortality: All humans are mortal.

Axiom GreeksHumanity: All Greeks are humans.
```

We introduce a graphical notation for axioms, where a horizontal bar is used to separate possible premises above the bar from the conclusion below the bar. Since a fact holds regardless of any premise, we display it as follows:

$$\frac{}{\texttt{All humans are mortal}}\text{[HumansMortality]}$$

Lemmas are affirmations that follow from all known facts. A lemma must be followed by a proof that demonstrates how it follows from known facts. We show this mechanism by stating the mortality of humans as a lemma, and prove it by simply applying the axiom *HumansMortality*.

Lemma 1. *The proposition* `All humans are mortal` *holds.*

We prove this lemma by simply invoking the axiom *HumansMortality*.

Proof.

$$\frac{}{\texttt{All humans are mortal}}\text{[HumansMortality]}$$

□

```
Lemma HumansMortality2: All humans are mortal.

Proof.
apply HumansMortality.
Qed.

End TraditionalLogic.
```

6 What do Axioms, Lemmas and Proofs Mean?

According to the discussion on semantics, we are free to fix a model $\mathcal{M}$, which selects a subset of our universe for each term. However, such a semantics may or may not meet the requirements posed by a given axiom. For example, if we choose $U^{\mathcal{M}} = \{0, 1\}$, $\texttt{humans}^{\mathcal{M}} = \{0\}$, and $\texttt{mortal}^{\mathcal{M}} = \{1\}$, then clearly the proposition

```
All humans are mortal
```

does not hold.

By asserting an axiom A, we are focusing our attention to only those models $\mathcal{M}$ for which $A^{\mathcal{M}} = T$. The lemmas that we prove while utilizing an axiom only hold in the models in which the axiom holds.

The ability of using an axiom in proofs comes at a price; the proof is valid only for those models in which the axiom holds. Thus, in a sense, our reasoning becomes weaker when we assert an axiom.

A proposition is called *valid*, if it holds in all models.

Exercise 1. *Is the proposition*

```
All humans are humans
```

valid? Use the semantics of categorical propositions (Section 3, Traditional Logic I) in your argument.

Exercise 2. *Is the proposition*

```
Some humans are humans
```

valid? Use the semantics of categorical propositions (Section 3, Traditional Logic I) in your argument.